Number base conversion is when you express a stored integer in an integer base, such as in octal (base 8) or binary (base 2). It also is involved when you take a string representing a number in a given base and convert it to the stored integer form. Normally, a stored integer is in binary, but that's typically invisible to the user, who normally enters or sees stored integers as decimal.



Task

Write a function (or identify the built-in function) which is passed a non-negative integer to convert, and another integer representing the base.

It should return a string containing the digits of the resulting number, without leading zeros except for the number   0   itself.

For the digits beyond 9, one should use the lowercase English alphabet, where the digit   a = 9+1,   b = a+1,   etc.

For example:   the decimal number   26   expressed in base   16   would be   1a.

Write a second function which is passed a string and an integer base, and it returns an integer representing that string interpreted in that base.

The programs may be limited by the word size or other such constraint of a given language. There is no need to do error checking for negatives, bases less than 2, or inappropriate digits.




begin
    % returns with numberInBase set to the number n converted to a string in %
    % the specified base. Number must be non-negative and base must be in    %
    % range 2 to 36                                                          %
    procedure convertToBase( integer    value  n
                           ; integer    value  base
                           ; string(32) result numberInBase
                           ) ;
    begin
        string(36) baseDigits;
        integer    val, strPos;
 
        assert( n >= 0 and base >= 2 and base <= 36 );
 
        baseDigits    := "0123456789abcdefghijklmnopqrstuvwxyz";
        numberInBase  := " ";
        val           := n;
        strPos        := 31;
        while
            begin
                % a(b//c) is the substring of a starting at b with length c. %
                % The first character is at position 0. The length must be   %
                % an integer literal so it is known at compile time.         %
                numberInBase( strPos // 1 ) := baseDigits( val rem base // 1 );
                val    := val div base;
                strPos := strPos - 1;
                val > 0
            end
        do begin end
    end convertToBase ;
 
    % returns the string numberInBase converted to an integer assuming       %
    % numberInBase ia a string in the specified base                         %
    % base must be in range 2 to 36, invalid digits will cause the program   %
    % to crash, spaces are ignored                                           %
    integer procedure convertFromBase( string(32) value numberInBase
                                     ; integer    value base
                                     ) ;
    begin
        string(36) baseDigits;
        integer    val, cPos;
 
        assert( base >= 2 and base <= 36 );
 
        baseDigits    := "0123456789abcdefghijklmnopqrstuvwxyz";
        val           := 0;
        for strPos := 0 until 31 do begin
            string(1) c;
            c := numberInBase( strPos // 1 );
            if c not = " " then begin
                cPos := 0;
                while baseDigits( cPos // 1 ) not = c do cPos := cPos + 1;
                val  := ( val * base ) + cPos;
            end
        end;
        val
    end convertFromBase ;
 
    % test the procedures                                                    %
    string(32) baseNumber;
    i_w := 3; % set integer output width                                     %
    for i := 2 until 36 do begin
        convertToBase( 35, i, baseNumber );
        write( 35, i, baseNumber, " ", convertFromBase( baseNumber, i ) );
    end
end.